Chapter 1

Getting Started

Before starting, please read Appendix A so that your Maple interface is properly configured. This chapter introduces some of the basic Maple commands associated with assigning labels to numbers and formulas and creating plots.

1.1 Maple as a Calculator

A Maple input line is indicated by an input prompt > at the left hand margin. A Maple statement (or command) is entered (or executed) by typing it on an input line with a semicolon (;) at the end and then pressing the ⟨ENTER⟩ or ⟨RETURN⟩ key. Try entering

```
> 2 + 5;
```

7

(Don't type the > prompt since this is provided by the computer.) Maple's output, 7, is displayed in the center of the next line.

If a command is entered and Maple replies

```
> 2 + 5

Warning, premature end of input
```

then you have forgotten the semicolon. Click at the end of the line, type a semicolon and press ⟨ENTER⟩ again. (See Section 11.1.)

If a command or an expression is entered incorrectly then click the mouse on the line, edit it and then re-execute it by pressing ⟨ENTER⟩ again.

Maple can do arithmetic on formulas entered on an input line. The standard arithmetic operations are

`+`	addition	`-`	subtraction
`*`	multiplication	`/`	division
`^`	exponentiation		

The standard order of operations is exponentiation before multiplication and division and then addition and subtraction. Within a level, commands are done

left to right. To be safe, use parentheses to be sure that the operations are performed in the desired order. For example, `(3+4)/7;` is not the same as `3+4/7;`. Be sure to use round parentheses `( )` rather than square brackets `[ ]` or curly braces { }, which have other meanings in Maple.

CAUTION: A typical mistake is typing

```
> (2+4)(3-1);
```

$$6$$

instead of

```
> (2+4)*(3-1);
```

$$12$$

Maple will not multiply without the `*` sign. Rather it will give very peculiar results without warning. Try it! (See Section 11.1.)

Maple knows a large number of standard mathematical functions including:

square root	`sqrt`
absolute value	`abs`
natural exponential	`exp`
natural logarithm	`ln`
trig functions	`sin, cos, tan, sec, csc, cot`
inverse trig functions	`arcsin, arccos, arctan,` `arcsec, arccsc, arccot`

It is important to understand the distinction between numbers that Maple knows exactly, such as 2, 1/3, $\sqrt{2}$, and π, and *floating point decimal numbers* such as 2.0, .33333, 1.414, and 3.14. The number 1/3 is an expression that represents the exact value of one-third, whereas .33333 is a floating-point decimal approximation of 1/3.

If numbers are entered as integers, then Maple normally returns exact answers. For example, enter

```
> (1+3)/6;
```

$$\frac{2}{3}$$

Maple returns 2/3 rather than its decimal approximation .6666666667. To get the decimal approximation, use the `evalf` command. For example, try

```
> evalf(22/79+34/23);
```

$$1.756741882$$

Notice that parentheses are needed around the expression. Here, `evalf` means to evaluate the expression as a floating-point decimal number. Alternatively, one of the numbers can be entered as a decimal. For example, type

```
> 22./79 + 34/23;
```

$$1.756741883$$

and Maple returns the answer in decimal form.

1.2 Assigning Variables

Maple answers are often used again in subsequent calculations and therefore Maple provides a way to store and recall earlier results. One way to refer to an earlier result is to use the percent sign `%`, which refers to the immediately preceding result. For example, the calculation of 2^6+1 can be done in two steps to show intermediate answers. The result of the first operation is given to the second part as `%`.

```
> 2^6; %+1;
```

$$64$$
$$65$$

Note that the input line contains two Maple statements (since there are two semicolons) and therefore there are two Maple outputs.

Names or labels can also be used to store and refer to results. For example, the number 22./79+34/23 can be assigned to the variable `a` by typing

```
> a:=22./79+34/23;
```

$$a := 1.756741883$$

Maple commands of this form are called assignment statements and the := sign indicates that the quantity on the right is to be assigned to the variable name on the left. Now, the number 1.756741883 can be recalled by typing `a`.

EXAMPLE 1: Compute $(1.756741883)^2$, $\dfrac{1}{1.756741883}$, and $\sqrt{1.756741883}$.

```
> a^2;
```

$$3.086142043$$

```
> 1/a;
```

$$0.5692355887$$

```
> sqrt(a);
```

$$1.325421398$$

To more easily distinguish between various labels, use descriptive names.

EXAMPLE 2: Enter an expression that describes the area of a circle of radius r

```
> Area:=Pi*r^2;
```

$$Area := \pi\, r^2$$

Note that π is entered with an upper case P. With a lower case p, Maple will show the Greek letter π but won't recognize its mathematical meaning. To evaluate this area when $r = 5$, enter `r:=5;`. The value $r = 5$ will automatically be substituted into *Area*.

```
> r:=5; Area; evalf(%);
```

$$r := 5$$
$$25\,\pi$$
$$78.53981635$$

Note, however,

```
> evalf(pi*r^2);
```

$$25.\,\pi$$

has the wrong type of π. Also note that labels are case sensitive: the label `Area` is different from the label `area`.

EXAMPLE 3: Compute the profit on an item if its retail price is \$4.95 and its wholesale cost is \$2.80.

SOLUTION: Enter the price and cost:

```
> Price:=4.95;
```

$$Price := 4.95$$

```
> Cost:=2.80;
```

$$Cost := 2.80$$

Then the profit is given by

```
> Profit:=Price-Cost;
```

$$Profit := 2.15$$

NOTE: A variable keeps its value until it is assigned a new value or until it is cleared (or unassigned) or until Maple is restarted. For example, the value of the variable `Price` can be unassigned by assigning it to its name in single quotes:

```
> Price:='Price';
```

$$Price := Price$$

Alternatively, the values of the variables `Cost` and `Profit` can be unassigned by using the `unassign` command:

```
> unassign('Cost','Profit');
```

Now the variables `Price`, `Cost` and `Profit` have no values assigned to them.

More generally, the command

```
> restart;
```

will clear out everything from Maple's memory, thus unassigning all variables.

TIP: In fact it is probably useful to execute a `restart` at the beginning of each new homework problem.

The same thing happens if you save your worksheet (by selecting FILE > SAVE or clicking on the save icon) and reopen it at a later date. Everything is on the screen but nothing is in memory. To re-execute the whole worksheet, select EDIT > EXECUTE > WORKSHEET or click on the *!!!* icon. Remember the worksheet is re-executed in order from top to bottom. So if you typed or modified your worksheet out of order, it may not re-execute properly.

1.3 Algebra Commands

We have seen how to manipulate numbers and assign them to variables (or labels). Maple can also manipulate algebraic expressions involving variables.

For example, to multiply out the expression $(3x-2)^2(x^3+2x)$, type

```
> (3*x-2)^2*(x^3+2*x); expand(%);
```

$$(3\,x-2)^2\,(x^3+2\,x)$$

$$9\,x^5+22\,x^3-12\,x^4-24\,x^2+8\,x$$

As mentioned earlier, the percent `%` refers to the output preceding the percent, in this case, the expression $(3x-2)^2(x^3+2x)$.

NOTE: It is easy to make a typing error when entering a complicated expression, such as $(3x-2)^2(x^3+2x)$. To prevent such errors from affecting a Maple command (such as `expand`), first type the expression without the command and press ⟨ENTER⟩ as follows:

```
> (3*x-2)^2*(x^3+2*x);
```

$$(3\,x-2)^2\,(x^3+2\,x)$$

Examine Maple's output to make sure that the expression is entered correctly. Then click the mouse back at the end of the previous line and add the Maple command `expand(%)` as done above. Putting `expand(%)` on the same line also guarantees that if you change the expression to expand, the `expand(%)` will also be re-executed.

To factor the polynomial x^6-1, type

```
> x^6-1; factor(%);
```

$$x^6-1$$

$$(x-1)\,(x+1)\,(x^2+x+1)\,(x^2-x+1)$$

Another useful command is `simplify`. For example, to simplify the expression $\dfrac{x^2-x}{x^3-x}-\dfrac{x^2-1}{x^2+x}$ enter

```
> (x^2-x)/(x^3-x)-(x^2-1)/(x^2+x); simplify(%);
```

$$\frac{x^2-x}{x^3-x}-\frac{x^2-1}{x^2+x}$$

$$-\frac{x^2-x-1}{x\,(x+1)}$$

The following command will also simplify the expression.

```
> simplify((x^2-x)/(x^3-x)-(x^2-1)/(x^2+x));
```

$$-\frac{x^2-x-1}{x\,(x+1)}$$

However, with this syntax, it is harder to keep track of the parentheses in such a long expression. In addition, this command does not display the original expression and therefore it cannot be checked for typing errors.

NOTE: Maple has an on-line Help facility that is invoked by typing `?` followed by the command. For example, to get help with the `factor` command, type (No semicolon is necessary.)

```
> ?factor
```

Alternatively, if the command word is already typed in the worksheet, just click on it and press F2. Also explore the Help Browser and the Search facilities by clicking on HELP > MAPLE HELP.

1.4 Plots

The `plot` command is best introduced with an example. To plot the graph of

$$f = \frac{x^2 - 4}{x + 1}$$

over the interval $-6 \leq x \leq 6$, type

```
> f:=(x^2-4)/(x+1);
```

$$f := \frac{x^2 - 4}{x + 1}$$

```
> plot(f, x=-6..6);
```



The scale on the y-axis is much different from the scale on the x-axis because of the very large function values when x is close to -1 (where the function becomes undefined). To get a more reasonable plot, the y-range should be specified. For example, to view the piece of the graph with $-10 \leq y \leq 10$, enter

```
> plot(f, x=-6..6, y=-10..10);
```



Notice the vertical line at the vertical asymptote $x = -1$. This is an artifact of the way Maple makes plots. It calculates points and connects them with straight lines. To see these points, right-click in the plot and select STYLE > POINT. Try that above. Alternatively, include the plot option `style=point` as follows:

```
> plot(f, x=-6..6, y=-10..10, style=point);
```



By default, Maple plots 49 points. To increase the number of points, add the option `numpoints=#` (but don't make the number absurdly large):

```
> plot(f, x=-6..6, y=-10..10, style=point, numpoints=201);
```



To connect the dots, while eliminating the vertical line, include the option `discont=true`:

```
> plot(f, x=-6..6, y=-10..10, discont=true);
```



Next notice that the scale on the y-axis is different from the scale on the x-axis. Normally, Maple adjusts the scales of both axes so that the plot fills the plot window. To equalize scales on the x- and y-axes, right-click in the plot and select SCALING CONSTRAINED. Try that above. Alternatively, include the option `scaling=constrained`

```
> plot(f, x=-6..6, y=-10..10, discont=true, scaling=constrained);
```



Other options to `plot` will be discussed in Chapter 2. A complete list of the plot options may be seen by executing

```
> ?plot,options
```

By changing the plot range, different aspects of the graph can be viewed. For example, the above plot shows the x-intercepts at -2 and 2, the y-intercept at -4 and the vertical asymptote at $x = -1$. Changing the x- and y-ranges to `x=-200..200, y=-200..200` displays the graph for larger values of x:

```
> plot(f, x=-200..200, y=-200..200, discont=true);
```



Notice the graph of the function approaches the line $y = x - 1$ which is a slant asymptote. (You will learn later how to compute slant asymptotes.) However, with such large values of x, the vertical asymptote at $x = -1$ becomes obscured.

If the x-range is omitted from the plot command, Maple will plot the expression over the interval $-10 \leq x \leq 10$. (In other words, this is the default range).

More than one expression can be graphed on the same plot by enclosing several expressions using curly braces { } or square brackets [ ]. For example, to add the skewed asymptote $y = x - 1$ to the above plot, type

```
> g:=x-1;
```

$$g := x - 1$$

```
> plot({f,g}, x=-6..6, y=-10..10, discont=true);
```



CAUTION: Opening too many plots on your computer may degrade its performance. It is wise to close unneeded plots by clicking on the plot and pressing ⟨CTRL-DELETE⟩. Likewise, to reduce the size of saved files, click on the EDIT menu and select REMOVE OUTPUT > FROM WORKSHEET before saving. To recover the output, click on the EDIT menu and select EXECUTE > WORKSHEET or click on the toolbar icon which looks like three exclamation points, *!!!*.

1.5 Summary

- Maple does everything that a graphing calculator will do.
- Maple function arguments must be in parentheses; for example, $\sin x$ must be written as `sin(x)`.
- Maple executes arithmetic commands in a predefined order of precedence. Use parentheses to modify the order. Use care in entering expressions and examine output for correctness.
- Multiplication requires an asterisk. Juxtaposition of symbols is not allowed in Maple as a synonym for multiplication: `(x+4)(x+2)` is viewed as a function evaluation, not a product.
- Most calculators store numbers only as floating-point decimals, with a preselected number of digits of accuracy. Maple has an alternative exact mode that treats the fraction 1/3 not as a decimal 0.3333333333 to any number of 3's, but instead as a list of two integers, 1 and 3.

- The distinction between storage modes is important because Maple will often not be able to express an answer in terms of a list of integers, and so it will parrot back the original expression typed in. This does not represent a syntax error: Maple simply doesn't know how to give an exact answer for `sin(1)`.
- Exact answers can be converted to floating-point decimal approximations of any number of digits by using `evalf`.
- It is unnecessary (and unwise) to retype intermediate results in subsequent calculations or even to copy and paste them. It is unwise because if you correct a previous mistake or even just change something, the copied result will not automatically update. Maple provides two alternatives: the assignment command, `:=`, in which a label is associated with a given output; and the percent, `%`.
- Once a number or algebraic expression is assigned a label, any statement that contains that label treats it as a synonym for the number or expression itself.
- Once an assignment has been made, Maple remembers that assignment until it is told otherwise. A label may be unassigned by reassigning it to its own name enclosed in single forward quotes or by including it in an `unassign` command or by executing a `restart`. One common source of frustration is forgetting that a label already has an assigned value when trying to use it as a free variable. (See Section 11.4.)
- Save your worksheet frequently. When you reopen a worksheet, re-execute it by clicking on the *!!!* icon.
- Algebraic expressions can be manipulated with the commands `expand`, `factor`, and `simplify`.
- Use `?`*`command`* to get help on a command including proper syntax and examples. The command name does not have to be exact to get an answer. Also, use the Help Browser and Search facility under the HELP menu.
- Maple can be used to create graphs of expressions. Know how to use Maple to plot a graph, how to modify the domain and range, and how to specify options. Be aware that the defaults do not always show important features of the graph and know how to change the defaults to do so.
- Clicking the mouse on a graph shows the approximate coordinates of the point on the top left corner of the Maple window.

1.6 Exercises

1. Assign the variable name `a` to the number $2\pi/12$. Evaluate each of the following exactly and then use `evalf(%)` to compute the decimal approx-

imations:
a^2, $1/a$, $\sqrt{a}$, $a^{1.3}$, $\sin(a)$, $\tan(a)$, and $\tan^{-1}(a)$.

2. The number of significant digits can be temporarily changed from the default value of 10 by modifying the `evalf` command. For example, to calculate a^2 to 20 significant digits, type `evalf(a^2,20);` Repeat Exercise 1 with 20 significant digits using `evalf(%,20)`.

3. The number of significant digits can be permanently changed from the default value of 10 to some other number, such as 25, with the command `Digits:=25`, Repeat Exercise 1 with 25 significant digits. Remember to return the number of digits to 10 by executing `Digits:=10;`.

4. Display the following expressions and then `expand` them.

   (a) $(x^2 - 4x + 3)^3(x^2 + 4)$

   (b) $(x - a)^6$

   NOTE: If you have done Exercise 1, 2 or 3, the label a already has a value assigned to it. Recall that this value should be unassigned by typing `a:='a';` before part b..

5. Display and `factor` the expression $x^2 - \frac{5}{3}x - \frac{2}{3}$. What happens if this expression is changed to $x^2 - \frac{5.0}{3}x - \frac{2.0}{3}$?

6. Try factoring $x^2 - 4x - 9$. What happens? Repeat with the command `factor(%,sqrt(13))` .

7. Factor $x^8 - 1$ (over the integers). Repeat with each of the following as the second argument of `factor`, and discuss the differences.

   (a) `sqrt(2)`

   (b) `I`

   (c) {`sqrt(2),I`}

   (d) `real`

   (e) `complex`

8. Simplify

$$\frac{8x^2}{x^4 - 1} - \frac{4}{x^2 - 1}$$

   What is the difference in the domains of the original expression and its simplified form?

9. Plot the graph of $\sec(x)$. Experiment with the x- and y-ranges to obtain a reasonable plot of one period of $\sec(x)$. Eliminate the vertical asymptote.

10. (a) Plot the graph of $\dfrac{4x^2-2x+2}{x-1}$ over a small interval containing $x=1$, for example, $0 \le x \le 2$. Experiment with the y-range to obtain a reasonable plot. What happens to the graph near $x=1$? Add the option `discont=true`.

    (b) Now plot the same expression over a large interval such as $-100 \le x \le 100$ without `discont=true`. Note that the behavior of the graph near $x=1$ is no longer apparent. Why do you think this happens? Try adding `style=point`. Now add `discont=true`, both with and without `style=point`.

11. Repeat Exercise 10a for $\dfrac{4x^2-2x-2}{x-1}$. To explain the result try simplifying the expression. (This is an example of a function which is undefined at $x=1$ but whose limit is defined there.)

12. Plot the expressions $\sin(x)$, $\sin(2x)$, and $\sin(4x)$ over the interval $0 \le x \le 2\pi$ on the same coordinate axes. Now plot the same expressions over the interval $0 \le x \le 4\pi$. At what numbers are they all equal?

13. Use `evalf(Pi,40);` to give the first 40 digits of π.

14. Compute the exact and floating-point values of $\sin(\pi/3)$ and $\sin(3)$.

15. Compute the number of seconds in one year, showing the units in your product as each factor is entered, e.g. `365*day/year`.

16. Just as `factor` will factor a polynomial, there is also an `ifactor` (integer factor) command that gives the prime decomposition of an integer. Use `ifactor` to show that $2^{21}-1$ is not prime. How long would it have taken you to find the factors by hand?

17. Compute 27!. Recall: $4! = 4*3*2*1$. How many final zeros are there? Factor 27! using `ifactor`. How many factors of 5 are there? Why is this not a coincidence? NOTE: Maple knows the `!` sign.

18. What happens when the command `expand` is applied to $(a-2b)/c$? Practice the technique of entering the expression and checking its Maple output to see that it is entered correctly. Then go back and add `expand(%)` on the same line.

19. Apply `expand` to $\ln\left(\dfrac{a}{bc}\right)$. Repeat this but preceded by `assume(a>0, b>0, c>0);` which tells Maple a, b and c are positive. Why did this make a difference? To remove the assumptions, execute `a:='a';` and similarly for `b` and `c`. NOTE: The tildes (~) mean there are assumptions on the variables. See Appendix A to eliminate the tildes.

20. Factor the expression $e^{2x}-1$, by first using `expand`, then `factor`. Compare this with just using `factor`.

21. Load the `student` package by typing `with(student);` Label the points $P_0 := [1, 3]$ and $P_1 := [4, 7]$ on the same input line. (NOTE: Subscripts are entered by putting them in square brackets, e.g. `P[0]`.) Find out the syntax for determining the slope between the two points via `?slope`. Find the slope between P_0 and P_1 using the `slope` command. Assign the value that you find to the variable m. Find the equation of the line through P_0 and P_1.

22. Plot the expressions $-x/2 + 3/2$ and $-3x + 4$ on the same graph with x-values between 0 and 2. Right click in the plot and select SCALING CONSTRAINED to avoid distortion in the plot. Click on the intersection with the mouse to find the coordinates of the intersection of the lines $y = -x/2 + 5/2$ and $y = -3x + 5$. The coordinates are shown at the top-left of the Maple window just under the FILE menu.